```
>
```

Boucle, test et procedure

```
> restart;
```

Les nombres premiers de 100 à 2 (en ordre décroissant) :

```
> l:=[];
  for i from 100 to 2 by -1 do
         if isprime(i) then
                 l:=[op(l),i]
         fi;
  od;
  l;
```

$$l := [\ ]$$

$$[97, 89, 83, 79, 73, 71, 67, 61, 59, 53, 47, 43, 41, 37, 31, 29, 23, 19, 17, 13, 11, 7, 5, 3, 2]$$

Les 30 premiers nombres premiers :

```
> l:=[];n:=0;i:=2;
  while nops(l)<30 do
         if isprime(i) then
                 l:=[op(l),i];
         fi;
         i:=i+1
  od:
  l;
```

$$l := [\ ]$$

$$n := 0$$

$$i := 2$$

$$[2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97, 101, 103, 107, 109, 113]$$

Orthonormalisation, par la méthode de Schmidt, d'une famille libre e=[e[1],...], pour un produit scalaire ps ::
procedure non récursive :

```
> ortho1:=proc(e,ps)
  local k,n,u,i,c,v;
  k:=nops(e);
  n:=v->sqrt(ps(v,v));
  u:=[e[1]/n(e[1])];
  for i from 2 to k do
          c:=[seq(ps(u[j],e[i]),j=1..i-1)];
          v:=e[i]-sum('c[j]*u[j]','j'=1..i-1);
          u:=[op(u),v/n(v)]
  od;
  return u;
  end:
> e:=[1,t,t^2,t^3]:ps:=(p,q)->int(exp(-t)*p*q,t=0..infinity):or
  tho1(e,ps);
```

$$\left[1,\ t-1,\ \frac{1}{2}t^2+1-2t,\ \frac{1}{6}t^3-1+3t-\frac{3}{2}t^2\right]$$

procedure récursive :

```
> ortho2:=proc(e,ps)
  local k,n,ee,u,c,v;
  k:=nops(e);
  n:=v->sqrt(ps(v,v));
  if k=1 then
         return [e[1]/n(e[1])]
  else
         ee:=e[1..k-1];
         u:=ortho2(ee,ps);
         c:=[seq(ps(u[j],e[k]),j=1..k-1)];
         v:=e[k]-sum('c[j]*u[j]','j'=1..k-1);
         return [op(u),v/n(v)]
  fi;
  end:
> ortho2(e,ps);
```

$$\left[1,\ t-1,\ \frac{1}{2}t^2+1-2t,\ \frac{1}{6}t^3-1+3t-\frac{3}{2}t^2\right]$$

Nombres entiers

```
> restart;
> assume(n,integer):sin(n*Pi),cos(n*Pi);
```

$$0,\ (-1)^{n\sim}$$

```
> n:='n':sin(n*Pi),cos(n*Pi);
```

$$\sin(n\,\pi),\ \cos(n\,\pi)$$

```
> ifactor(58136);isprime(58136);[type(58136,odd),type(58136,eve
  n)];
```

$$(2)^3\ (13)^2\ (43)$$

$$\text{false}$$

$$[\text{false},\ \text{true}]$$

```
> iquo(47,7);irem(47,7);[floor(47/7),round(47/7),trunc(47/7)];
```

$$6$$

$$5$$

$$[6,\ 7,\ 6]$$

Polynômes et fractions rationnelles

```
> restart;
> A:=36*x^3+87*x^2+24*x-12;A2:=x^2+x*y+y^2+x-y+1;
```

$$A := 36\,x^3+87\,x^2+24\,x-12$$

$$A2 := x^2+x\,y+y^2+x-y+1$$

```
> F:=factor(A);expand(F);
```

$$F := 3\,(3\,x+2)\,(x+2)\,(4\,x-1)$$

$$36\,x^3+87\,x^2+24\,x-12$$

```
> B:=x^2+x+1:divEucl:=[quo(A,B,x),rem(A,B,x)];
```

$$divEucl := [\,36\,x+51,\,-63-63\,x\,]$$

```
> P:=A*(x^2+1):factor(P);factor(P,I);factor(P,complex);
```

$$3\,(3\,x+2)\,(x+2)\,(4\,x-1)\,(x^2+1)$$

$$-3\,(4\,x-1)\,(x+2)\,(-x+I)\,(x+I)\,(3\,x+2)$$

$$36.\,(x+2.)\,(x+0.6666666667)\,(x+1.\,I)\,(x-1.\,I)\,(x-0.2500000000)$$

```
> coeffs(A,x);coeff(A,x,2);
```

$$-12,\,36,\,87,\,24$$

$$87$$

```
> collect(A2,x);collect(A2,y);
```

$$x^2+(1+y)\,x-y+1+y^2$$

$$y^2+(-1+x)\,y+x^2+x+1$$

```
> N1:=x^4+x^2+1:N2:=x^3+x^2+x+1:convert(N1/N2,parfrac,x);convert(N1/N2,parfrac,x,I);convert(N1/N2,parfrac,x,complex);
```

$$-1+x+\frac{3}{2\,(x+1)}+\frac{1-x}{2\,(x^2+1)}$$

$$-1+x+\frac{3}{2\,(x+1)}-\frac{\frac{1}{4}-\frac{1}{4}\,I}{x+I}+\frac{\frac{1}{4}+\frac{1}{4}\,I}{-x+I}$$

$$-1+x-\frac{0.2500000000+0.2500000000\,I}{x-1.000000000\,I}+\frac{1.500000000}{x+1.}-\frac{0.2500000000-0.2500000000\,I}{x+1.000000000\,I}$$

Algèbre linéaire

```
> restart:with(LinearAlgebra):
```

Créer une matrice, un vecteur :

```
> a:=Matrix(2,3);
```

$$a := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> b:=Matrix([[0,1],[3,4]]);
```

$$b := \begin{bmatrix} 0 & 1 \\ 3 & 4 \end{bmatrix}$$

```
> tg:=(i,j)->i+j;c:=Matrix(2,3,tg);
```

$$tg := (i,j) \rightarrow i+j$$

$$c := \begin{bmatrix} 2 & 3 & 4 \\ 3 & 4 & 5 \end{bmatrix}$$

```
> d:=Matrix(3,shape=identity);
```

$$d := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> e:=DiagonalMatrix([seq(i^2,i=1..3)]);
```

$$e := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 4 & 0 \\ 0 & 0 & 9 \end{bmatrix}$$

```
> f:=Matrix(2,2,symbol=m);
  sys:=convert(f.b-b.f,set);
  subs(solve(sys),f);
```

$$f := \begin{bmatrix} m_{1,1} & m_{1,2} \\ m_{2,1} & m_{2,2} \end{bmatrix}$$

sys :=

$$\{3\,m_{1,2} - m_{2,1},\ m_{2,1} - 3\,m_{1,2},\ m_{1,1} + 4\,m_{1,2} - m_{2,2},\ 3\,m_{2,2} - 3\,m_{1,1} - 4\,m_{2,1}\}$$

$$\begin{bmatrix} m_{1,1} & m_{1,2} \\ 3\,m_{1,2} & m_{1,1} + 4\,m_{1,2} \end{bmatrix}$$

```
> u:=Vector([1,-4,-4]);
```

$$u := \begin{bmatrix} 1 \\ -4 \\ -4 \end{bmatrix}$$

```
> v:=Vector([2,5,3]):w:=Vector([-2,-2,0]):g:=Matrix([u,v,w]);
```

$$g := \begin{bmatrix} 1 & 2 & -2 \\ -4 & 5 & -2 \\ -4 & 3 & 0 \end{bmatrix}$$

```
> h:=DiagonalMatrix(w);
```

$$h := \begin{bmatrix} -2 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

Calculer :

```
> b^(-1).c.g;
```

$$\begin{bmatrix} \frac{71}{3} & \frac{-83}{3} & \frac{26}{3} \\ -26 & 31 & -10 \end{bmatrix}$$

```
> c.u;
```

$$\begin{bmatrix} -26 \\ -33 \end{bmatrix}$$

```
> Transpose(g).g-g.Transpose(g);
```

$$\begin{bmatrix} 24 & -40 & 4 \\ -40 & -7 & -45 \\ 4 & -45 & -17 \end{bmatrix}$$

```
> k:=t->t^2:map(k,c);
```

$$\begin{bmatrix} 4 & 9 & 16 \\ 9 & 16 & 25 \end{bmatrix}$$

Fonctions usuelles :

```
> Determinant(g);
```

$$6$$

```
> NullSpace(c);
```

$$\left\{\begin{bmatrix} 1 \\ -2 \\ 1 \end{bmatrix}\right\}$$

```
> l:=Matrix([[8,-1,-5],[-2,3,1],[4,-1,-1]]);
  CharacteristicPolynomial(l,x);
```

$$l := \begin{bmatrix} 8 & -1 & -5 \\ -2 & 3 & 1 \\ 4 & -1 & -1 \end{bmatrix}$$

$$-32 + x^3 - 10\,x^2 + 32\,x$$

```
> Eigenvalues(l);
```

$$\begin{bmatrix} 2 \\ 4 \\ 4 \end{bmatrix}$$

```
> Eigenvectors(l);
```

$$\begin{bmatrix} 4 \\ 4 \\ 2 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 1 \\ -1 & 0 & 1 \\ 1 & 0 & 1 \end{bmatrix}$$

```
> lReduc:=[JordanForm(l,output=[J,Q])];lP:=lReduc[2];lT:=lReduc
  [1];lP.lT.lP^(-1);
```

$$lReduc := \left[\begin{bmatrix} 2 & 0 & 0 \\ 0 & 4 & 1 \\ 0 & 0 & 4 \end{bmatrix}, \begin{bmatrix} \frac{-1}{2} & 3 & \frac{3}{2} \\ \frac{-1}{2} & -3 & \frac{1}{2} \\ \frac{-1}{2} & 3 & \frac{1}{2} \end{bmatrix}\right]$$

$$lP := \begin{bmatrix} \frac{-1}{2} & 3 & \frac{3}{2} \\ \frac{-1}{2} & -3 & \frac{1}{2} \\ \frac{-1}{2} & 3 & \frac{1}{2} \end{bmatrix}$$

$$lT := \begin{bmatrix} 2 & 0 & 0 \\ 0 & 4 & 1 \\ 0 & 0 & 4 \end{bmatrix}$$

$$\begin{bmatrix} 8 & -1 & -5 \\ -2 & 3 & 1 \\ 4 & -1 & -1 \end{bmatrix}$$

Pour une matrice symétrique réelle :

```
> s:=Matrix([[4,1,1],[1,4,1],[1,1,4]]);sReduc:=[JordanForm(s,ou
  tput=[J,Q])];
  sP:=sReduc[2];sPOrtho:=Matrix(GramSchmidt([seq(Column(sP,i),i
  =1..3)],normalized));
```

```
sT:=sReduc[1];
sPOrtho.sT.Transpose(sPOrtho);
```

$$s := \begin{bmatrix} 4 & 1 & 1 \\ 1 & 4 & 1 \\ 1 & 1 & 4 \end{bmatrix}$$

$$sReduc := \left[\begin{bmatrix} 3 & 0 & 0 \\ 0 & 6 & 0 \\ 0 & 0 & 3 \end{bmatrix}, \begin{bmatrix} \frac{-1}{3} & \frac{1}{3} & -1 \\ \frac{-1}{3} & \frac{1}{3} & 0 \\ \frac{2}{3} & \frac{1}{3} & 1 \end{bmatrix} \right]$$

$$sP := \begin{bmatrix} \frac{-1}{3} & \frac{1}{3} & -1 \\ \frac{-1}{3} & \frac{1}{3} & 0 \\ \frac{2}{3} & \frac{1}{3} & 1 \end{bmatrix}$$

$$sPOrtho := \begin{bmatrix} -\frac{\sqrt{6}}{6} & \frac{\sqrt{3}}{3} & -\frac{\sqrt{2}}{2} \\ -\frac{\sqrt{6}}{6} & \frac{\sqrt{3}}{3} & \frac{\sqrt{2}}{2} \\ \frac{\sqrt{6}}{3} & \frac{\sqrt{3}}{3} & 0 \end{bmatrix}$$

$$sT := \begin{bmatrix} 3 & 0 & 0 \\ 0 & 6 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

$$\begin{bmatrix} 4 & 1 & 1 \\ 1 & 4 & 1 \\ 1 & 1 & 4 \end{bmatrix}$$

Géométrie (1)

```
> restart:
```

Dessiner des lignes polygonales :

```
> a:=[0,1]:b:=[5,1]:c:=[4,6]:
  plot([a,b,c,a],scaling=constrained);d1:=%:
```




```
> m:=(a+b)/2:n:=(b+c)/2:p:=(c+a)/2:
  d2:=plot([m,c]):d3:=plot([n,a]):d4:=plot([p,b]):
  with(plots):display([d1,d2,d3,d4]);
```




```
> aa:=[op(a),0]:bb:=[op(b),0]:cc:=[op(c),0]:dd:=[2,4,3]:
> spacecurve([aa,bb,cc,aa,dd,bb,cc,dd]);
```



Chercher des longueurs, angles, aires,... :

```
> with(LinearAlgebra):Norm(Vector(c-m),2);theta:=arccos(DotProduct(Vector(c-m),Vector(b-a))/Norm(Vector(c-m))/Norm(Vector(b-a)));evalf(theta);
```

$$\frac{\sqrt{109}}{2}$$

$$\theta := \arccos\left(\frac{3}{10}\right)$$

$$1.266103673$$

```
> CrossProduct(Vector(bb-aa),Vector(cc-aa));
```

$$\begin{bmatrix} 0 \\ 0 \\ 25 \end{bmatrix}$$

Suites et séries

```
> restart;
```

Définir une suite, une série :

```
> x:=n->1/n!;y:=n->sin(n^2)/n;
```

$$x := n \rightarrow \frac{1}{n!}$$

$$y := n \rightarrow \frac{\sin(n^2)}{n}$$

```
> u:=n->(1+1/n)^n;
```

$$u := n \rightarrow \left(1+\frac{1}{n}\right)^n$$

```
> evalf([seq(u(10*i),i=1..20)],4);
```

[2.594, 2.653, 2.674, 2.685, 2.692, 2.696, 2.699, 2.701, 2.703, 2.705, 2.706, 2.707, 2.708, 2.709, 2.709, 2.710, 2.710, 2.711, 2.711, 2.712]

```
> n:=15:w:=[1]:for i from 1 to n do w:=[op(w),w[i]+1/i!]
  od:evalf(w,6);
```

[1., 2., 2.50000, 2.66667, 2.70833, 2.71667, 2.71806, 2.71825, 2.71828, 2.71828, 2.71828, 2.71828, 2.71828, 2.71828, 2.71828, 2.71828]

Etudier une suite :

```
> limit(u(x), x=infinity);
```

$$e$$

```
> l1:=[seq([i,u(i)],i=1..10)]:l2:=[seq([i,w[i]],i=1..10)]:
  plot([l1,l2],style=line,legend=["suite u","suite
  w"],view=[1..10,1.6..2.8]);
```



Etude d'une série :

```
> s:=n->sum('y(i)','i'=1..n);
```

$$s := n \rightarrow \sum_{'i'=1}^{n} 'y(i)'$$

```
> evalf([seq(s(10*p),p=10..20)]);
```

[0.1828229346, 0.1800839063, 0.1639744928, 0.1390325792, 0.1485106730, 0.1848960116, 0.1852600494, 0.1563707546, 0.1560054808, 0.1727260750, 0.1765953442]

```
> sum('x(i)','i'=0..infinity);
```

$$e$$

Fonctions de R dans R ou C

```
> restart;
```

Function vs expression :

```
> f1:=t->exp(-t)-t;evalf(f1(1));e1:=f1(t);
```

$$f1 := t \to e^{(-t)} - t$$
-0.6321205588
$$e1 := e^{(-t)} - t$$

```
> e2:=tan(t)-t;evalf(subs(t=5,e2));f2:=unapply(e2,t);
```

$$e2 := \tan(t) - t$$
-8.380515006
$$f2 := t \to \tan(t) - t$$

```
> f3:=x->piecewise(x<0,-2*x,x<1,x^2,1);
  f4:=f3@f3;g:=n->f3@@n;
```

$$f3 := x \to \text{piecewise}(x < 0, -2\,x, x < 1, x^2, 1)$$
$$f4 := f3^{(2)}$$
$$g := n \to f3^{(n)}$$

Résoudre une équation/inéquation :

```
> solve(e1);evalf(%);
```

LambertW(1)
0.5671432904

```
> fsolve(e2,t,Pi..3*Pi/2);s:=fsolve(e2,t,30..50);evalf(subs(t=s
  ,e2));
```

4.493409458
s := 39.24443236
$$-0.179\ 10^{-5}$$

Représenter une/des fonction/s :

```
> plot([e1,e2],t=0..1,legend=["e1","e2"],scaling=constrained);
```




```
> plot(e2,t=0..Pi,view=[0..Pi,-10..10],discont=true);
```




```
> plot([seq(g(n),n=1..4)],-.75..3,legend=[seq(convert(n,string)
  ,n=1..4)]);
```



Simplifier :

```
> a:=cos(3*x):expand(a);
```

$$4\cos(x)^3 - 3\cos(x)$$

```
> b:=sin(x)*sin(y):combine(b);
```

$$\frac{1}{2}\cos(x-y) - \frac{1}{2}\cos(x+y)$$

```
> simplify((cos(x))^2+(sin(x))^2);
```

$$1$$

```
> simplify(x^3+y^3,[x+y=S,x*y=P]);
```

$$-3\,S\,P + S^3$$

```
> rationalize((2+sqrt(3))^2/(1-sqrt(5)));radnormal(%);
```

$$-\frac{(2+\sqrt{3})^2\,(\sqrt{5}+1)}{4}$$

$$-\frac{7\sqrt{5}}{4}-\sqrt{5}\sqrt{3}-\frac{7}{4}-\sqrt{3}$$

```
> assume(t>0):sqrt(t^2);t:='t':sqrt(t^2);
```

$$t\sim$$

$$\sqrt{t^2}$$

Limite, DL, extrema :

```
> limit(e1,t=+infinity);
  limit(e1/t,t=+infinity);
```

$$-\infty$$

$$-1$$

```
> limit(e2,t=Pi/2,right);
```

$$-\infty$$

```
> dl:=series(e1,t=0,8);partRegul:=convert(dl,polynom);
  taylor(e1,t=0);
  asympt(f1(1/t),t);
```

$$dl := 1-2\,t+\frac{1}{2}t^2-\frac{1}{6}t^3+\frac{1}{24}t^4-\frac{1}{120}t^5+\frac{1}{720}t^6-\frac{1}{5040}t^7+O(t^8)$$

$$partRegul := 1-2\,t+\frac{1}{2}t^2-\frac{1}{6}t^3+\frac{1}{24}t^4-\frac{1}{120}t^5+\frac{1}{720}t^6-\frac{1}{5040}t^7$$

$$1-2\,t+\frac{1}{2}t^2-\frac{1}{6}t^3+\frac{1}{24}t^4-\frac{1}{120}t^5+O(t^6)$$

$$1-\frac{2}{t}+\frac{1}{2\,t^2}-\frac{1}{6\,t^3}+\frac{1}{24\,t^4}-\frac{1}{120\,t^5}+O\left(\frac{1}{t^6}\right)$$

```
> maximize(t^3-2*t+1,t=0..3,location);
  minimize(t^3-2*t+1,t=0..3);
```

$$22,\ \{[\{t=3\},22]\}$$

$$1-\frac{4\sqrt{2}\sqrt{3}}{9}$$

Dérivée, intégrale, primitive :

```
> diff(e2,t);
  l:=expand([seq(diff(e2,t$i),i=1..4)]);
```

$$\tan(t)^2$$

$$l := [\tan(t)^2,\ 2\tan(t)+2\tan(t)^3,\ 2+8\tan(t)^2+6\tan(t)^4,\ 16\tan(t)+40\tan(t)^3+24\tan(t)^5]$$

```
> D(f2);
  seq((D@@i)(f2),i=1..3);
```

$$t\rightarrow\tan(t)^2$$

$$t\rightarrow\tan(t)^2,\ t\rightarrow 2\tan(t)\,(1+\tan(t)^2),\ t\rightarrow 2\,(1+\tan(t)^2)^2+4\tan(t)^2\,(1+\tan(t)^2)$$

```
> Int(e1,t=0..1);int(e1,t=0..1);
  int(e2,t=0..Pi/2);
  int(sqrt(e2),t=0..Pi/2);evalf(%);
```

```
int((sin(t))^2/t^2,t=0..+infinity);
```

$$\int_0^1 e^{(-t)} - t \, dt$$

$$\frac{1}{2} - e^{(-1)}$$

$$\infty$$

$$\int_0^{\frac{\pi}{2}} \sqrt{\tan(t) - t} \, dt$$

$$1.552555813$$

$$\frac{\pi}{2}$$

```
> E2:=int(e2,t=0..x);assume(x>0,x<Pi/2):int(e2,t=0..x);EE2:=int
  (e2,t);
```

```
Warning, unable to determine if 1/2*Pi+Pi*_Z19 is between 0 and x; try to
use assumptions or set _EnvAllSolutions to true
```

$$E2 := \int_0^x \tan(t) - t \, dt$$

$$-\ln(\cos(x\sim)) - \frac{x\sim^2}{2}$$

$$EE2 := -\ln(\cos(t)) - \frac{t^2}{2}$$

```
> x:='x':E3:=int(f3,0..x);
```

$$E3 := \begin{cases} -x^2 & x \leq 0 \\ \frac{x^3}{3} & x \leq 1 \\ x - \frac{2}{3} & 1 < x \end{cases}$$

Equations différentielles

```
> restart;
```

Ecrire une EDO (ou un système différentiel) :

```
> ode1:=diff(y(t),t$2)=-y(t);
  ode2:=diff(y(t),t$2)+sin(y(t));
  ode3:=diff(y(t),t$2)+diff(y(t),t)+y(t);
  sys:=diff(x(t),t)=-y(t),diff(y(t),t)=x(t);
```

$$ode1 := \frac{d^2}{dt^2} y(t) = -y(t)$$

$$ode2 := \left(\frac{d^2}{dt^2} y(t)\right) + \sin(y(t))$$

$$ode3 := \left(\frac{d^2}{dt^2} y(t)\right) + \left(\frac{d}{dt} y(t)\right) + y(t)$$

$$sys := \frac{d}{dt} x(t) = -y(t), \frac{d}{dt} y(t) = x(t)$$

Ecrire une/des condition/s initiale/s :

```
> ic1:=y(0)=1,D(y)(0)=0;
  ic2:=y(0)=0,D(y)(0)=1;
  ics:=x(0)=1,y(0)=0;
```

$$ic1 := y(0) = 1, D(y)(0) = 0$$
$$ic2 := y(0) = 0, D(y)(0) = 1$$
$$ics := x(0) = 1, y(0) = 0$$

Résoudre :

```
> sol1:=dsolve(ode1,y(t));
  yy:=subs(sol1,y(t));
  yyy:=subs({_C1=1/2,_C2=sqrt(3)/2},yy);
```

$$sol1 := y(t) = \_C1 \sin(t) + \_C2 \cos(t)$$
$$yy := \_C1 \sin(t) + \_C2 \cos(t)$$
$$yyy := \frac{1}{2} \sin(t) + \frac{1}{2}\sqrt{3} \cos(t)$$

```
> sol1ic:=dsolve({ode1,ic1},y(t));
  zz:=subs(sol1ic,y(t));
  plot(zz,t=-Pi..Pi);
```

$$sol1ic := y(t) = \cos(t)$$
$$zz := \cos(t)$$




```
> sol2:=dsolve(ode2,y(t));
  sol3:=dsolve({ode2,ic1},y(t),numeric);
  sol3(1);
  op(2,op(2,sol3(1)));
```

sol2 :=

$$\int^{y(t)} \frac{1}{\sqrt{2\cos(\_a)+\_C1}}\, d\_a - t - \_C2 = 0,\ \int^{y(t)} -\frac{1}{\sqrt{2\cos(\_a)+\_C1}}\, d\_a - t - \_C2 = 0$$

sol3 := proc(x_rkf45) ... end proc

$$\left[t = 1.,\ y(t) = 0.600085309600074890,\ \frac{d}{dt}y(t) = -0.754963973883389538\right]$$

0.600085309600074890

```
> with(DEtools):
> DEplot(ode2,y(t),0..1,[[ic1],[ic2]]);
```




```
> sol3:=dsolve({ode3,ic1},y(t));
```

$$sol3 := y(t) = \frac{1}{3}\sqrt{3}\, e^{\left(-\frac{t}{2}\right)} \sin\left(\frac{\sqrt{3}\, t}{2}\right) + e^{\left(-\frac{t}{2}\right)} \cos\left(\frac{\sqrt{3}\, t}{2}\right)$$

```
> solsys:=dsolve({sys,ics},[x(t),y(t)]);
  courbe:=subs(solsys,[x(t),y(t)]);
  plot([op(courbe),t=0..Pi]);
  plot(courbe,t=0..Pi);
```

$$solsys := \{x(t) = \cos(t), y(t) = \sin(t)\}$$

$$courbe := [\cos(t), \sin(t)]$$


1
0.8
0.6
0.4
0.2
0
-1
-0.5
0.5
1



Fonctions d'une variable vectorielle

```
> restart;
```

Définir une fonction :

```
> e1:=x^2-3*x+y^2+3*y+3;f1:=unapply(e1,[x,y]);
```

$$e1 := x^2 - 3\,x + y^2 + 3\,y + 3$$

$$f1 := (x, y) \rightarrow x^2 - 3\,x + y^2 + 3\,y + 3$$

```
> f2:=(x,y)->x*(ln(x^2))+y^2;e2:=f2(x,y);
```

$$f2 := (x, y) \rightarrow x\,\ln(x^2) + y^2$$

$$e2 := x\,\ln(x^2) + y^2$$

Représenter une fonction :

```
> plot3d(e1,x=-5..5,y=-5..5);
```



Extrema :

```
> minimize(e1,x=-5..5,y=-5..5,location);
```

$$\frac{-3}{2}, \left\{\left[\left\{x=\frac{3}{2}, y=\frac{-3}{2}\right\}, \frac{-3}{2}\right]\right\}$$

Dérivée partielle, intégrale :

```
> s1:={diff(e1,x),diff(e1,y)};m1:=solve(s1);subs(m1,e1);
```

$$s1 := \{2\,x-3, 2\,y+3\}$$

$$m1 := \left\{x=\frac{3}{2}, y=\frac{-3}{2}\right\}$$

$$\frac{-3}{2}$$

```
> p:=diff(e2,x):q:=diff(e2,y):r:=diff(e2,x$2):s:=diff(e2,x,y):t
  :=diff(e2,y$2):
  m2:=allvalues(solve({p,q}));H:=Matrix([[r,s],[s,t]]);
```

$$m2 := \{x = \frac{e}{e^2}, y = 0\}, \{x = -\frac{e}{e^2}, y = 0\}$$

$$H := \begin{bmatrix} \frac{2}{x} & 0 \\ 0 & 2 \end{bmatrix}$$

```
> with(student):Doubleint(e1,x=-5..5,y=-5..5);value(%);
```

$$\int_{-5}^{5}\int_{-5}^{5} x^2 - 3x + y^2 + 3y + 3 \, dx \, dy$$

$$\frac{5900}{3}$$

Géométrie (2)

```
> restart:with(plots):
```

Dessiner des courbes :

```
> zc:=t*exp(I*t):complexplot(zc,t=-10..10);
```




```
> c1:=1+cos(t):polarplot(c1,t=-Pi..Pi,scaling=constrained);
```




```
> c2:=t+sin(t),1+cos(t):plot([c2,t=0..3*Pi],scaling=constrained
  );cc2:=%:
```




```
> c3:=x^2+x*y+4*y^2=6:
  d1:=implicitplot(c3,x=-3..3,y=-2..2,scaling=constrained,numpo
  ints=15,color=red):
  d2:=implicitplot(c3,x=-3..3,y=-2..2,scaling=constrained,numpo
  ints=150,color=blue):
  display([d1,d2]);
```




```
> c4:=cos(t),sin(t),t:spacecurve([c4,t=-3*Pi..3*Pi],numpoints=1
  00);
```



Dessiner des surfaces :

```
> s1:=3*x^2+2*y^2:plot3d(s1,x=-2..2,y=-2..2,view=-0..20);
```




```
> s2:=r*cos(t),r*sin(t),r^2:plot3d([s2],t=-Pi..Pi,r=0..2);ss2:=
  %:
```



```
> s3:=x^2+2*y^2-x*z-y*z=1:implicitplot3d(s3,x=-2..2,y=-2..2,z=-
  2..2,grid=[13,13,13]);
```



Ajouter une tangente, un plan tangent :

```
> a2:=subs(t=Pi/4,[c2]);t2:=subs(t=Pi/4,[diff(c2[1],t),diff(c2[2],t)]);
```

$$a2 := \left[\frac{\pi}{4} + \sin\left(\frac{\pi}{4}\right), 1 + \cos\left(\frac{\pi}{4}\right)\right]$$

$$t2 := \left[1 + \cos\left(\frac{\pi}{4}\right), -\sin\left(\frac{\pi}{4}\right)\right]$$

```
> tangente:=plot([a2-2*t2,a2+2*t2],color=blue):display([cc2,tangente]);
```




```
> b2:=subs([t=Pi/4,r=1],[s2]);
  der1:=subs([t=Pi/4,r=1],[seq(diff(s2[i],t),i=1..3)]);
  der2:=subs([t=Pi/4,r=1],[seq(diff(s2[i],r),i=1..3)]);
  p:=b2+[seq(a*der1[i]+b*der2[i],i=1..3)];
```

$$b2 := \left[\cos\left(\frac{\pi}{4}\right), \sin\left(\frac{\pi}{4}\right), 1\right]$$

$$der1 := \left[-\sin\left(\frac{\pi}{4}\right), \cos\left(\frac{\pi}{4}\right), 0\right]$$

$$der2 := \left[\cos\left(\frac{\pi}{4}\right), \sin\left(\frac{\pi}{4}\right), 2\right]$$

$$p := \left[-\frac{a\sqrt{2}}{2} + \frac{b\sqrt{2}}{2} + \frac{\sqrt{2}}{2}, \frac{a\sqrt{2}}{2} + \frac{b\sqrt{2}}{2} + \frac{\sqrt{2}}{2}, 2\,b + 1\right]$$

```
> planT:=plot3d(p,a=-1..1,b=-1..1):display([ss2,planT]);
```



>